

Solutions

To derive the maximum benefit from these exercises, make an honest effort to create the *Mathematica* code before consulting the solution offered here.

Part I. The *Mathematica* Notebook

■ Section 1. Cell structure: Input/Output

2.* Simplify the square root of 19,220 by entering and executing

Sqrt[19220]

Sqrt[19220]

$62\sqrt{5}$

Then obtain the integer factorization of 19220 by entering and executing

FactorInteger[19220]

FactorInteger[19220]

$\{\{2, 2\}, \{5, 1\}, \{31, 2\}\}$

The output means that $19220 = 2^2 * 5 * 31^2$

3.* Obtain the integral of the expression $e^{3x} \cos(2x) \sin(4x)$. You may enter it into the **Integrate** function as follows

Integrate[Exp[3x] Cos[2x] Sin[4x], x]

Integrate[Exp[3x] Cos[2x] Sin[4x], x]

$\frac{1}{390} e^{3x} (-30 \cos[2x] - 26 \cos[6x] + 45 \sin[2x] + 13 \sin[6x])$

■ Section 2. Mixing Text and Mathematics

16.* Save your Notebook (Command-S) then print it.

See the pdf file entitled My2ndNotebook.nb.

Part II. Calculations and Calculus with Mathematica

■ Section 1. Getting Started: *Mathematica* as a Calculator

6. Use the **Table** function to make the following lists.

a. { 2, 4, 6, 8, 10, 12 }

b. { 20, 40, 60, ... , 260 }

c. The 30 prime numbers starting with 11 and ending with 139. Hint. Use **Prime[n]** whose output is the nth prime number. Name the list P.

7.* Add the numbers in the sequence P found in 6 c. Hint. Execute **Sum[P[[k]], {k,30}]** .

```
P = Table[ Prime[n], {n,5,33} ]  
{11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67,  
71, 73, 79, 83, 89, 97, 101, 103, 107, 109, 113, 127, 131, 137}  
  
Sum[ P[[k]], {k,29} ]  
1971
```

9. Make the list Cu of the cubes of the integers 2, 5, 6, 9, 12, 44. Hint: Enter the integers in a list named L, Then enter $Cu = L^3$. Add the numbers in the list of cubes and then display the prime factorization of the sum. What do you notice about the prime factorization? (Hint. The integer 87990 is called "square free".)

10.* Obtain the prime factorization of the product of the integers in the list Cu of cubes described in Exercise 9.

```
L = {2,5,6,9,12,44}  
Cu = L^3  
{2, 5, 6, 9, 12, 44}  
{8, 125, 216, 729, 1728, 85184}  
  
Product[ Cu[[k]], {k,6} ]  
FactorInteger[%]  
23178378313728000  
{{2, 18}, {3, 12}, {5, 3}, {11, 3}}
```

11. Use the **Factor** function to factor the following polynomial expressions. (Note. Begin by entering $x = .$ to free the x variable.)

a. $x^3 - x^2 + x - 1$

b. $x^7 - x^6 + x^5 - \dots - 1$ Hint. Enter this as $P = -\text{Sum}[(-x)^k, \{k, 0, 7\}]$ then execute **Factor[P]**.

12.* Use **Solve** to obtain the zeros of the polynomials in Exercise 11.

$$P = -\text{Sum}[(-x)^k, \{k, 0, 3\}]$$

$$-1 + x - x^2 + x^3$$

$$Z = \text{Solve}[P == 0, x]$$

$$\{\{x \rightarrow -i\}, \{x \rightarrow i\}, \{x \rightarrow 1\}\}$$

13.* Add the zeros of the polynomial in 11 a and the zeros of the polynomial in 11 b. Hint. For example, if P is the polynomial in 11b, enter $Z = \text{Solve}[P==0, x]$ and then $\text{Sum}[x/Z[[k]], \{k, 3\}]$

$$\text{Sum}[x/Z[[k]], \{k, 3\}]$$

$$1$$

$$\text{Solve}[\text{Sum}[(-x)^k, \{k, 0, 7\}] == 0, x]$$

$$\text{Sum}[x/.\%, \{k, 7\}]$$

$$\{\{x \rightarrow -i\}, \{x \rightarrow i\}, \{x \rightarrow 1\}, \{x \rightarrow -(-1)^{1/4}\},$$

$$\{x \rightarrow (-1)^{1/4}\}, \{x \rightarrow -(-1)^{3/4}\}, \{x \rightarrow (-1)^{3/4}\}\}$$

$$1$$

■ Section 2. Symbolics: Equations and Assignments


4.* Plot the expression $x^3 - x^2 + 0.005x + \cos(x) - 0.7$ with the entries $y = x^3 - x^2 + 0.05x + \text{Cos}[x] - 0.7$ and

$$\text{Plot}[y, \{x, -2, 2\}]$$

$$y = x^3 - x^2 + 0.05x + \text{Cos}[x] - 0.7$$

$$\text{Plot}[y, \{x, -2, 2\}]$$

$$-0.7 + 0.05x - x^2 + x^3 + \text{Cos}[x]$$



- Graphics -

Then enter **FindRoot**[$y==0, \{x, 0\}$] to see which zero **FindRoot** finds.

```
FindRoot[y==0, {x,0}]
```

```
{x → -0.386373}
```

Using the graph as a guide, obtain an approximation to the largest positive zero using **FindRoot**.

```
FindRoot[y==0, {x,1,2}]
```

```
{x → 1.18978}
```

6. Graph the function $y = \cos(x^2) - \sin(2x)$ over the interval from $x = 0$ to $x = 2$. Name the Plot **Gy**. Find the derivative function and call it **yp** using $yp = D[y,x]$. Find the root of **yp** near $x = 1.5$. Name it **xmin**. Use the following entry to plot the graph of **y** and the low point.

```
Show[ Gy, ListPlot[ {{x,y}/.xmin}, PlotStyle->PointSize[0.02] ] ]
```


Start everything with **Clear[x,y]**.

7.* Find the area of the region between the graph of **y** in Exercise 6 and the x -axis. Hint: Numerically integrate the absolute value of **y** from $x = 0$ to $x = 2$ via the entry **NIntegrate[Abs[y], {x,0,2}]**.

```
y = Cos[x^2] - Sin[2x]
```

```
Plot[ y, {x,0,2}]
```

```
Cos[x^2] - Sin[2 x]
```



```
NIntegrate[ Abs[y], {x,0,2}]
```

```
0.91477
```

■ Section 3. Functions as Transformations

Functions play a key role in many applications of mathematics. *Mathematica* makes functions in a natural way. The **Table** function can be used to make tables of data. Use **MatrixForm** to display the data in an array.

0. Pull down the **Help** menu, choose **Help Browser...**, Choose **The Mathematica Book/A Practical Introduction to Mathematica/Functions and Programs**. Read the Help pages listed in the last column.

Open a new Maple worksheet and do the following

1. Define the function $f(x) = \cos(x) - x \tan(x)$ Then use the entry **f' [x]** to obtain the derivative formula.

2. Continuing 1. Plot f over the interval $x = -1$ to $x = 1$ using `Plot[f[x], {x,-1,1}]`. Then find the area of the region below the graph of f and above the x axis. Hint. This will require the values b where $f(b) = 0$. Find the positive value using `b = FindRoot[f[x]==0, {x,0,1}][[1,2]]`.

Hint. By symmetry, the negative zero is at $x = -b$. Check this is true by computing $f(-b)$.

3. Continuing 2. Plot the graph of the function f from -1 to 1 and the tangent line segment to the graph at the point $x = 0.5$, $y = f(0.5)$ over the interval $x = 0$ to $x = 1$. Hint. Define the tangent line function using `T[x_] := f[0.5] + f'[0.5](x - 0.5)`. Then execute

```
Show[ Plot[ f[x], {x,-1,1}], Plot[ T[x], {x,0,1} ]
```

Now jazz up the plot by making the curve red and the tangent line blue. (Just edit the Show entry.)

4. Continuing 3. Find the length of the curve plotted in Exercise 2. Hint. Do the integration numerically using

```
NIntegrate[ Sqrt[1 + f'[x]^2], {x,-1,1} ]
```


5.* An animation. The tangent line plot in Exercise 3 can be animated as follows. First clear `T` and `a` with `Clear[T,a]` and define the function `T(a,x)` whose value at (a,x) is the formula for the tangent line to the graph of f at $(a,f(a))$: `T[a_,x_] := f[a] + f'[a](x - a)`. Then load the Animation package and apply `Animate` as shown below.

```
<<Graphics`Animation`
Animate[ Show[ Plot[ f[x], {x,-1,1}, PlotStyle->RGBColor[1,0,0]],
             Plot[ T[a,x], {x,a-0.5,a+0.5}, PlotStyle->RGBColor[0,0,1]],
             PlotRange->{{-1,1},{-2,2}}
          ], {a,-1,1} ]
```

Mathematica will make 24 plots over the specified range of a values. Once this is done, collapse the 24 output cells into one by double clicking on the single blue bracket that encloses them all. Then select the collapsed bracket (it will have a down arrow indicating that there are cells collapsed inside), pull down the **Cell** menu, and choose **Animate Selected Graphics**. This should show the plots in sequence. Animation controls will appear at the bottom left of the Notebook window.

```
f[x_] := Cos[x] - x Tan[x]
T[a_,x_] := f[a] + f'[a](x-a)
```

```
<<Graphics`Animation`
Animate[ Show[ Plot[ f[x], {x,-1,1}, PlotStyle->RGBColor[1,0,0]],
             Plot[ T[a,x], {x,a-0.5,a+0.5}, PlotStyle->RGBColor[0,0,1]],
             PlotRange->{{-1,1},{-2,2}}
          ], {a,-1,1} ]
```



Read the Help page for animated graphics. Type 1.9.11 into the search field.

6. Solving a differential equation. First enter **Clear[x,y]**. The **DSolve** function solves differential equations. The syntax is

DSolve[DE, y[x], x]

where DE is a differential equation for y(t) (or the name of one). Define a simple first order differential equation as follows


DE = y'[x] + x y[x] == x

Obtain the general solution to DE using the **DSolve** function as above. Then obtain the solution satisfying $y(0) = 0$ using

soln = DSolve({DE, y[0]==0}, y[x], x)

7.* Plot the solution in Exercise 6 using **Plot[y[x]/.soln, {x,-2,2}]**

```
Clear[x,y]  
soln = DSolve[ {y'[x] + x y[x] == x, y[0]==0}, y[x], x]  
Plot[ y[x]/.soln, {x,-2,2}]  
{ {Y[x] → e-x2/2 (-1 + ex2/2) } }
```



8.* Use **Table** and **MatrixForm** to make a display of the values of the second solution in Exercise 6 at $x = 0, 0.2, 0.4, \dots, 1.0$. Use

Table[y[x]/.soln, {x,0,1,0.2}]/MatrixForm

```
Table[ y[x]/.soln, {x,0,1,0.2}]/MatrixForm  

$$\begin{pmatrix} 0 \\ 0.0198013 \\ 0.0768837 \\ 0.16473 \\ 0.273851 \\ 0.393469 \end{pmatrix}$$

```

Now make an array displaying the values x and y(x) with the following entry

Table[{x,y[x]}/.soln[[1]], {x,0,1,0.2}]/MatrixForm

```
Table[ {x,y[x]}/.soln[[1]], {x,0,1,0.2}]/MatrixForm
```

```
( 0      0
 0.2  0.0198013
 0.4  0.0768837
 0.6  0.16473
 0.8  0.273851
 1.   0.393469)
```

Part III. First Order Ordinary Differential Equations


■ Section 1. Entering, Solving, Plotting

7.* Enter the differential equation $y' + y = \cos(t)$ with the name DE. Obtain the general solution.

- Plot some solutions starting at points evenly spaced on the y axis.
- Use plot some solutions starting at points evenly spaced on the t axis.
- Plot solutions starting at points evenly spaced around the unit circle in the style of the two plots on page 35 of the manual. That is, one picture runs time forward, another runs time backward.

Hint. Use **Table** to generate a list of solutions, then plot them. See the example below.

```
DE = y'[t] + y[t] == Cos[t];
solna = Table[ DSolve[{DE,y[0]==y0},y[t],t], {y0,-2,2}];
Show[ ListPlot[ Table[{0,k},{k,-2,2}], PlotStyle->PointSize[0.02]],
      Plot[ Evaluate[Table[y[t]/.solna[[k]},{k,5}]], {t,0,12}], AspectRatio->
2/3]
```




Solution to b.

```

solnb = Table[ DSolve[{DE,y[t0]==0},y[t],t], {t0,-2,2}];
Show[ ListPlot[ Table[{k,0},{k,-2,2}], PlotStyle->PointSize[0.02]],
      Table[Plot[ y[t]/.solnb[[k]], {t,k-3,12}],{k,5}], AspectRatio->1/3 ]

```




Solutions to c.


```

solnc1 = Table[ DSolve[{DE,y[Cos[k*Pi/6]]==Sin[k*Pi/6]},y[t],t], {k,12}];
Show[ ListPlot[ Table[{Cos[k*Pi/6],Sin[k*Pi/6]},{k,12}],
      PlotStyle->PointSize[0.02]],
      Table[Plot[ y[t]/.solnc1[[k]], {t,Cos[k*Pi/6],2}],{k,12}] ,
      PlotRange->{{-2,2},{-2,2}}, AspectRatio->1/1]

```




```
solnc2 = Table[ DSolve[{DE,y[Cos[k*Pi/6]]==Sin[k*Pi/6]},y[t],t], {k,12}];
Show[ ListPlot[ Table[{Cos[k*Pi/6],Sin[k*Pi/6]},{k,12}],
      PlotStyle->PointSize[0.02]],
      Table[Plot[ y[t]/.solnc1[[k]], {t,-2,Cos[k*Pi/6]},{k,12}],
      PlotRange->{{-2,2},{-2,2}}, AspectRatio->1/1]
```



■ Section 2. Working with Solutions: Modeling

3.* The following problem is adapted from Ledder, Chapter 1, Section 1. Suppose you borrow \$12,000 to buy a car. The loan is to be paid in 60 equal monthly installments at an interest rate of 5% per year.

- a. Assume the payments are actually made continuously at whatever rate is needed to pay off the loan in 60 months. Determine the continuous rate per month that would be required.

Let $y(t)$ be the amount owed at time t and Q be the yearly rate of pay back.

```
DE = y'[t] == 0.05 y[t] - Q
soln = DSolve[ {DE, y[0]==12000} , y[t], t]
y'[t] == -Q + 0.05 y[t]
{{y[t] -> 2.71828^-0.05 t (12000. 7.38906^0.05 t - 20. 7.38906^0.05 t Q + 20. e^0.05 t Q)}}
```

```
Simplify[y[t]/.soln]
{e^0.05 t (12000. - 20. Q) + 20. Q}
```

```
Qrate = Solve[ (y[t]/.soln/.t->5)==0, Q]
{{Q -> 2712.49}}
```

```
MonthlyRate = (Q/.Qrate)/12
{226.041}
```

- b. Compare the answer to part a to the answer if 60 equal monthly payments are made at a constant annual interest rate of 5% applied to the outstanding balance. In other words, the first payment, due one month after the loan is made, would be

$$12000 \left(e^{\frac{0.05}{12}} - 1 \right) + P[1]$$

where $P[1]$ is the amount that is put towards reducing the principal (the %12,000) in the first month. The second payment is

$$(12000 - P[1]) \left(e^{\frac{0.05}{12}} - 1 \right) + P[2]$$

where $P[2]$ is the amount that is put towards reducing the principal in the second month. The n th payment is

$$(12000 - \sum_{k=1}^{n-1} P[k]) \left(e^{\frac{0.05}{12}} - 1 \right) + P[n]$$

Hint for b. The second payment is supposed to equal to the first. Therefore, the following equation must be satisfied.

```
equation = 12000(Exp[0.05/12]-1)+P[1] ==
           (12000-P[1])*(Exp[0.05/12]-1) + P[2]
50.1043 + P[1] == 0.00417536 (12000 - P[1]) + P[2]
```

This determines $P[2]$ in terms of $P[1]$.

```
Solve[ equation, P[2] ]
{{P[2] -> -1. (-50.1043 + 0.00417536 (12000. - 1. P[1]) - 1. P[1])}}
```

```
P[2] = P[2]/.%[[1]]//FullSimplify
0. + 1.00418 P[1]
```

So, we do this over and over again to define $P[3]$, $P[4]$, ... , $P[60]$ in terms of $P[1]$. See the Help page for the **Do** function.

```
Do[
eqn[n] = (12000 - Sum[P[k], {k, n-2}]) (Exp[0.05/12]-1)+P[n-1] ==
          (12000 - Sum[P[k], {k, n-1}]) (Exp[0.05/12]-1)+P[n];
P[n] = FullSimplify[P[n]/.Solve[eqn[n], P[n]][[1]]],
{n, 3, 60} ]
```

Find $P[1]$ using the fact that

$$\sum_{k=1}^{60} P[k] = 12000$$

```
Solve[Sum[P[k], {k, 60}] == 12000, P[1]]
{{P[1] -> 176.408}}
```

The first payment will be this amount plus the interest on 12000 (for one month).

```
12000(Exp[0.05/12]-1) + P[1]/.%
{226.512}
```

■ Section 3. Slope Fields


```
<<Graphics`PlotField`
```

2.* Use **PlotVectorField** and to make a nice-looking slope field for the autonomous equation $y' = \sin(y)$. (Use the window $-6 \leq t \leq 6$, $-6 \leq y \leq 6$).

```

u = 1; v = Sin[y];
vf = PlotVectorField[ {u,v}/Sqrt[u^2+v^2], {t,-6,6}, {y,-6,6}, Axes->True]

```




- a. Put some solution curves into the plot using **DSolve** and **Table**. Comment on the relationship between one curve and the next. Instead of **DSolve** the solution curves are obtained using *Mathematica's* numeric solver function, **NDSolve**. This function is discussed in the next section.

```

solns = Table[ NDSolve[{y'[t]==Sin[y[t]],y[0]==k},y[t],{t,-6,6}], {k,-6,6}
];
Show[vf, Table[ Plot[y[t]/.solns[[k]], {t,-6,6},
PlotStyle->RGBColor[1,0,0], {k,13} ] ]

```



- b. Obtain the general solution formula.

```

DSolve[ y'[t]==Sin[y[t]], y[t], t]
{{y[t] -> 2 ArcCot[ e^{-t-C[1]} ]}}

```

- c. Use **DSolve** to obtain the solution satisfying the initial condition $y(0) = 1$. What is the value of this solution when $t = 1$? Get the exact value and an approximation.

```
soln = DSolve[ {y'[t]==Sin[y[t]], y[0]==1}, y[t], t]
```

```
{ {y[t] → 2 ArcCot[e-t Cot[ $\frac{1}{2}$ ]] } }
```

```
y[t]/.soln/.t->1
```

```
N[%]
```

```
{ 2 ArcCot[ $\frac{\text{Cot}[\frac{1}{2}]}{e}$ ] }
```

```
{ 1.95629 }
```

d. Comment on the long-term behavior of solutions to this differential equation.

The direction field suggests that as t approaches infinity, all solutions have a horizontal asymptote. That is, the solutions have a finite limit (depending on the initial value) as t approaches infinity. The ArcCot function in the solution formula confirms this observation. One can also appeal to the fact that the values $y = k\pi$ are stationary points on the phase line. All solutions are either attracted to $k\pi$ or repelled from $k\pi$ for some integer k .

■ Section 4. Approximate Solutions

7.* Modify the ModEuler procedure to make a modular function that implements the classical Runge-Kutta algorithm (see Ledder, Section 2.6). Test in by applying it to the problems described in Exercises 2 - 6. (Copy and Paste). Is the error cut in half in problem 4?

```
RK[f_,t0_,y0_,h_,N_] :=
Module[
  {T,K1,K2,K3,K4,y},
  T[0] = t0; Y[0] = y0;
  T[n_] := T[n] = T[n-1] + h;
  K1[n_] := K1[n] = f[T[n-1],Y[n-1]];
  K2[n_] := K2[n] = f[T[n-1] + 0.5h,Y[n-1] + 0.5K1[n]*h];
  K3[n_] := K3[n] = f[T[n-1] + 0.5h,Y[n-1] + 0.5K2[n]*h];
  K4[n_] := K4[n] = f[T[n],Y[n-1] + K3[n]*h];
  Y[n_] := Y[n] = Y[n-1] + 1/6*(K1[n]+2K2[n]+2K3[n]+K4[n])h;
  Table[ {T[n],Y[n]}, {n,0,N} ]
]
```

Test the algorithm on the IVP in the manual (Exercise 5).

```
DE = y'[t] == Cos[t]y[t]
```

```
f[t_,y_] := Cos[t]y
```

```
y'[t] == Cos[t] y[t]
```

```
g = DSolve[ {DE,y[0]==1}, y, t][[1,1,2]];
```


```
g[t]
```

```
esin[t]
```

```

approx = RK[f,0,1,0.5,20];
pts = ListPlot[ approx, PlotStyle->PointSize[0.01], PlotRange->{0,3} ];
lines = ListPlot[ approx, PlotJoined->True, PlotStyle->RGBColor[1,0,0],
PlotRange->{0,3} ];
Show[ pts, lines, Plot[ g[t], {t,0,10} ],
AspectRatio->1/3, PlotLabel->"h = 0.5, Runge-Kutta" ]

```



Part IV. Linear Differential Equations

■ Section 1. Linear Oscillators

2.* Consider now the following damped system. Obtain the solution.

$$y'' + y' + 4y = 0, \quad y(0) = 2, \quad y'(0) = -3$$

From the form of the solution decide if the system is underdamped, critically damped, or overdamped.

$$\text{DE} = y''[t] + y'[t] + 4y[t] == 0$$

$$\text{soln} = \text{DSolve}[\{\text{DE}, y[0]==2, y'[0]==-3\}, y, t]$$

$$4 y[t] + y'[t] + y''[t] == 0$$

$$\left\{ \left\{ y \rightarrow \text{Function} \left[\{t\}, \frac{2}{15} e^{-t/2} \left(15 \cos \left[\frac{\sqrt{15} t}{2} \right] - 2 \sqrt{15} \sin \left[\frac{\sqrt{15} t}{2} \right] \right) \right] \right\} \right\}$$

The system is underdamped.

a. What is the pseudo-period of the oscillations?

The pseudo-period is

$$N[2\text{Pi}/(\text{Sqrt}[15]/2)] * \text{time_units}$$

$$3.24462 \text{ time_units}$$

b. What is the time constant?


The time constant is 2 time units.

c. Based upon your answer to part b estimate the time interval required for the oscillations to disappear from view.

In 10 time units the oscillations will decrease by a factor of e^5 which is about 150. They will disappear.


d. Plot the solution curve over the interval you named in part c.

```
Plot[ y[t]/.soln, {t,0,10}, PlotRange->{-1.5,2}]
```



e. Add to the curve in part d the curves defined by $A e^{-t/2}$ and $-A e^{-t/2}$ where $A = \sqrt{4 + \frac{16}{15}}$. Make them blue. What is the significance of these curves? Where did the formula for A come from?

```
A = Sqrt[4 + 16/15];
Plot[ {y[t]/.soln,A Exp[-t/2],-A Exp[-t/2]}, {t,0,10}, PlotRange->{-2.2,2.2}, PlotStyle->{RGBColor[1,0,0],RGBColor[0,0,1],RGBColor[0,0,1]}]
```




The blue curves envelope the solution curve. The constant A is derived from the standard formula used to convert the sum of a sine and a cosine into a sinusoid.

f. Convert the solution into the function g . Use g to plot the phase plane trajectory.

```
g = soln[[1,1,2]];
g[t]
```


$$\frac{2}{15} e^{-t/2} \left(15 \cos\left[\frac{\sqrt{15} t}{2}\right] - 2 \sqrt{15} \sin\left[\frac{\sqrt{15} t}{2}\right] \right)$$

```
traj = ParametricPlot[ {g[t],g'[t]}, {t,0,10}, PlotRange->{{-2,2},{-4,2}},
AspectRatio->1/1]
```



g. Add to the trajectory the points corresponding to $t = 0, 0.25, 0.5, 0.75, \dots, 2.0$.

```
Show[ ListPlot[ Table[{g[t],g'[t]}, {t,0,2,0.25}],
PlotStyle->PointSize[0.02]],
traj, PlotRange->{{-2,2},{-4,2}}, AspectRatio->1/1]
```



Section 2. State Space

2.* Damp the system slightly by changing the IVP to the following

$$y'' + 0.1y' + 4y = \cos(1.8 t) , y(0) = 2 , y'(0) = -3$$

Obtain the solution, call it soln and convert it into a function h.

```

DE = y''[t] + 0.1y'[t] + 4y[t] == Cos[1.8t]
soln = DSolve[ {DE,y[0]==2,y'[0]==-3},y,t];
h = soln[[1,1,2]];
N[h[t]]//FullSimplify
4 y[t] + 0.1 y'[t] + y''[t] == Cos[1.8 t]
(1.2459 - 5.55112 × 10-17 i) Cos[1.8 t] + (0.295082 + 2.77556 × 10-17 i) Sin[1.8 t] +
e-0.05 t ((0.754098 + 5.55112 × 10-17 i) Cos[1.99937 t] -
(1.74727 - 1.11022 × 10-16 i) Sin[1.99937 t])

```


- a. Based upon the solution formula determine the time constant for the beats. That is, how long will it take (approximately) for the beats to disappear from the solution curve as it settles down to its steady-state mode?

The time constant for the transient solution (when the beats will appear) is $\frac{1}{0.05} = 20$ time units.

- b. Plot the solution curve to verify your answer to part a.

The beats should disappear in about 100 time units.


```
Plot[ h[t], {t,0,100}, AspectRatio->1/4]
```




- c. Use h to obtain the phase plane and state space trajectories for this system.


```
ParametricPlot[ {h[t],h'[t]}, {t,0,100}, PlotLabel->"Phase Plane Trajectory", AspectRatio->3/2]
```

Phase Plane Trajectory




```
ParametricPlot3D[ {h[t],h'[t],t}, {t,0,100}, PlotLabel->"State Space Trajectory", AspectRatio->1/1, PlotPoints->400, ViewPoint->{2.664, -1.611, 10.327} ]
```




4.* Obtain a numeric solution for the IVP in Exercise 2. Sketch the time series for position and the time series for velocity. Then plot the phase plane trajectory and the state space trajectory.

```
soln = NDSolve[ {DE,y[0]==2,y'[0]==-3},y,{t,0,100}];
h = soln[[1,1,2]];
```


```
Plot[ h[t], {t,0,100}, AspectRatio->1/3, PlotLabel->"Time Series, Position"]
```




```
Plot[ h'[t], {t,0,100}, AspectRatio->1/3, PlotLabel->"Time Series, Velocity"]
```



```
ParametricPlot[ {h[t],h'[t]}, {t,0,100}, PlotLabel->"Phase Plane Trajectory", AspectRatio->3/2]
```



```
ParametricPlot3D[ {h[t],h'[t],t}, {t,0,100}, PlotLabel->"State Space Trajectory", AspectRatio->1/1, PlotPoints->400, ViewPoint->{2.664, -1.611, 10.327} ]
```



Section 3. Two Dimensional Systems

```
<<Graphics`PlotField`
```

4.* The model

$$x' = x(1 - y - x/a)$$

$$y' = y(1 - x - y/b)$$

is used to study competing species. See Ledder, Section 5.5, Exercise 11. Use **PlotVectorField** and **DSolve** to do the following.

- a. Draw the direction field when $1/a = 1.9$ and $1/b = 1.5$. Use the window $0 \leq x \leq 1, 0 \leq y \leq 1$.

```


a = 1/1.9; b = 1/1.5;
u = x(1-y-x/a); v = y(1-x-y/b);
vf= PlotVectorField[ {u,v}/Sqrt[u^2+v^2], {x,0,1}, {y,0,1}, Axes->True]

```

Power::infty : Infinite expression $\frac{1}{0}$ encountered. More...

∞::indet : Indeterminate expression 0 ComplexInfinity encountered. More...

∞::indet : Indeterminate expression 0 ComplexInfinity encountered. More...



- Graphics -

- b. Find the stationary point and add the nullclines to the direction field. Discuss the solutions based upon the picture you see.

```

statpt = Solve[ {u==0,v==0}, {x,y}]
{{x -> 0., y -> 0.}, {x -> 0., y -> 0.666667},
 {x -> 0.27027, y -> 0.486486}, {x -> 0.526316, y -> 0.}}


SP = {x,y}/.statpt[[3]]
{0.27027, 0.486486}

```

```

<<Graphics`ImplicitPlot`
nullclines = ImplicitPlot[{u==0,v==0}, {x,0,1}, {y,0,1}];
stationarypoint = ListPlot[ {SP}, PlotStyle->PointSize[0.025]];
flow = Show[ nullclines, stationarypoint, vf]

```



Based on this picture, all solution trajectories in the first quadrant will move towards the stationary point.


- c. Add solution curves corresponding to the following set of initial conditions (a circle of points around the stationary point) $x(0) = 0.3 + 0.2 \cos(k\pi/6)$, $y(0) = 0.5 + 0.2 \sin(k\pi/6)$, $k = 1, 2, \dots, 12$

```

solns = Table[
  NDSolve[ {x'[t]==x[t](1-y[t]-x[t]/a),
           y'[t]==y[t](1-x[t]-y[t]/b),
           x[0]==0.3+0.2Cos[k*Pi/6],y[0]==0.5+0.2Sin[k*Pi/6]},
           {x[t],y[t]}, {t,-10,10}], {k,11}];

```

```
Show[ flow,
Table[ ParametricPlot[ {x[t],y[t]}/.solns[[k]], {t,-10,10},
PlotStyle->{RGBColor[1,0,0],Thickness[0.01]} ], {k,11}],
PlotRange->{{0,1},{0,1}}]
```



Section 4. Matrix Methods

2.* Define the matrix B having the columns $\{0,1,-1\}$, $\{1,1,0\}$, $\{-1,0,1\}$. Hint.

```
B = Transpose[ { {0,1,-1}, {1,1,0}, {-1,0,1} } ]
```

```
B = Transpose[ { {0,1,-1}, {1,1,0}, {-1,0,1} } ];
B//MatrixForm
```

$$\begin{pmatrix} 0 & 1 & -1 \\ 1 & 1 & 0 \\ -1 & 0 & 1 \end{pmatrix}$$

a. Find the characteristic polynomial of B and factor it.

```
CharacteristicPolynomial[B,t]
Factor[%]
```

$$-2 + t + 2t^2 - t^3$$

$$-(-2 + t) (-1 + t) (1 + t)$$

b. Find B's eigenvalues with **Eigenvalues[B]**. Name them lambda.

```
lambda = Eigenvalues[B]
{2, -1, 1}
```

c. Find B's eigenvectors using **Eigenvectors[B]**. Name them V and define P to be the *transpose* of V.

```

V = Eigenvectors[B]
P = Transpose[V]
P//MatrixForm
{{-1, -1, 1}, {2, -1, 1}, {0, 1, 1}}
{{-1, 2, 0}, {-1, -1, 1}, {1, 1, 1}}

$$\begin{pmatrix} -1 & 2 & 0 \\ -1 & -1 & 1 \\ 1 & 1 & 1 \end{pmatrix}$$


```

d. Calculate $P^{-1}BP$

```

Inverse[P].B.P//MatrixForm

$$\begin{pmatrix} 2 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$


```

3.* Continuing 2. Obtain the solution to $\mathbf{v}' = \mathbf{B}\mathbf{v}$ satisfying $\mathbf{v}(0) = \{1,2,3\}$.

a. Do it first using **DSolve** applied to the system of linear differential equations defined by $\mathbf{v}' = \mathbf{B}\mathbf{v}$ with the appropriate initial conditions.

```

v[t_] := {x[t],y[t],z[t]}
DEsys = Table[ v'[t][[k]] == (B.v[t])[ [k]], {k,3}]
{x'[t] == y[t] - z[t], y'[t] == x[t] + y[t], z'[t] == -x[t] + z[t]}
soln1 = DSolve[ Join[DEsys,{x[0]==1,y[0]==2,z[0]==3}], {x,y,z}, t]
{{x -> Function[{t}, e-t],
  y -> Function[{t},  $\frac{1}{2} e^{-t} (-1 + 5 e^{2t})$ ], z -> Function[{t},  $\frac{1}{2} e^{-t} (1 + 5 e^{2t})$ ]}]}

```

Check:

```

DEsys/.soln1//FullSimplify
{{True, True, True}}

```

b. Do it second by making a fundamental matrix solution $X(t)$ defined as the matrix with the eigenvector solutions in the columns then computing

$$\mathbf{v}(t) = X(t) X(0)^{-1} \mathbf{v}(0)$$

```

X = Transpose[Table[ Exp[lambda[[k]]t]*V[[k]], {k,3}]]
{{-e2t, 2 e-t, 0}, {-e2t, -e-t, et}, {e2t, e-t, et}}
vsoln = X.Inverse[X/.t->0].{1,2,3};
vsoln//FullSimplify
{e-t, 2 Cosh[t] + 3 Sinh[t], 3 Cosh[t] + 2 Sinh[t]}

```

That is interesting output. Let's check that the solution displayed above is the same as the solution we got in part a.

```
{x[t],y[t],z[t]}/.soln1//FullSimplify
{{e^-t, 2 Cosh[t] + 3 Sinh[t], 3 Cosh[t] + 2 Sinh[t]}}
```

Indeed, it is.

c. Do it third by using the Matrix Exponential, Mexp. Once you have it, the solution is

$$v(t) = \text{Mexp } v(0)$$

```
Mexp = MatrixExp[B*t];
Mexp//MatrixForm

$$\begin{pmatrix} \frac{1}{3} e^{-t} (2 + e^{3t}) & \frac{1}{3} e^{-t} (-1 + e^{3t}) & -\frac{1}{3} e^{-t} (-1 + e^{3t}) \\ \frac{1}{3} e^{-t} (-1 + e^{3t}) & \frac{1}{6} e^{-t} (1 + 3 e^{2t} + 2 e^{3t}) & -\frac{1}{6} e^{-t} (-1 + e^t)^2 (1 + 2 e^t) \\ -\frac{1}{3} e^{-t} (-1 + e^{3t}) & -\frac{1}{6} e^{-t} (-1 + e^t)^2 (1 + 2 e^t) & \frac{1}{6} e^{-t} (1 + 3 e^{2t} + 2 e^{3t}) \end{pmatrix}$$

vsoln2 = Mexp.{1,2,3};
vsoln2//FullSimplify
{e^-t, 2 Cosh[t] + 3 Sinh[t], 3 Cosh[t] + 2 Sinh[t]}
```


Section 5. The Laplace Transform

5.* Consider the following initial value problem

$$y'' + y' + 9y = \text{UnitStep}(t - 2\pi) - \text{UnitStep}(t - 5\pi) + 2 \text{DiracDelta}(t - 9\pi), \quad y(0) = 0, \quad y'(0) = 1$$


a. Obtain the solution using **DSolve**. Plot it for $0 \leq t \leq 40$ and explain the behavior of the solution curve.

```
DE = y''[t] + y'[t] + 9y[t] == UnitStep[t-2Pi] - UnitStep[t-5Pi] +
2DiracDelta[t-9Pi]
soln = DSolve[ {DE,y[0]==0,y'[0]==1}, y, t];
9 y[t] + y'[t] + y''[t] == 2 DiracDelta[-9 pi + t] - UnitStep[-5 pi + t] + UnitStep[-2 pi + t]
Plot[ y[t]/.soln, {t,0,40}, PlotRange->{-0.4,0.6}, AspectRatio->1/3]
```



The solution curve shows the response to the steady force from $t = 6$ to $t = 16$ (roughly). See the plot of the driver below. Then there is no force until the kick provided by the Dirac delta at about $t = 27$. Note that at $t = 27$ the system reacts as it did at $t = 0$ except it goes twice as far from equilibrium.


```
Plot[ DE[[2]], {t,0,40}, PlotRange->{-0.1,1.4}, PlotLabel->"Unit Step Driver", AspectRatio->1/3]
```



- b. Obtain the solution using the method of Laplace transforms. How does the solution formula compare to the formula obtained in part a?

The Laplace transform method:

Transform the DE to TDE (in the s-domain)

```
TDE = LaplaceTransform[DE, t, s]
```

$$9 \text{LaplaceTransform}[y[t], t, s] + s \text{LaplaceTransform}[y[t], t, s] + s^2 \text{LaplaceTransform}[y[t], t, s] - y[0] - s y[0] - y'[0] == 2 e^{-9 \pi s} - \frac{e^{-5 \pi s}}{s} + \frac{e^{-2 \pi s}}{s}$$

Substitute the initial values and introduce Y for the transform of the solution.

```
TDE/.{y[0]->0,y'[0]->1,LaplaceTransform[y[t],t,s]->Y}
```

$$-1 + 9 Y + s Y + s^2 Y == 2 e^{-9 \pi s} - \frac{e^{-5 \pi s}}{s} + \frac{e^{-2 \pi s}}{s}$$

Solve for Y

```
Y = Y/.Simplify[Solve[%,{Y}]][[1]]
```

$$\frac{e^{-9 \pi s} (-e^{4 \pi s} + e^{7 \pi s} + 2 s + e^{9 \pi s} s)}{s (9 + s + s^2)}$$

Invert to get the Laplace transform solution.

```
Lsoln = InverseLaplaceTransform[Y,s,t]
```

$$\frac{2 e^{-t/2} \sin\left[\frac{\sqrt{35} t}{2}\right]}{\sqrt{35}} + \frac{4 e^{\frac{1}{2}(9 \pi - t)} \sin\left[\frac{1}{2} \sqrt{35} (-9 \pi + t)\right] \text{UnitStep}[-9 \pi + t]}{\sqrt{35}} - \left(\frac{1}{9} - \frac{1}{315} e^{\frac{1}{2}(5 \pi - t)} \left(35 \cos\left[\frac{1}{2} \sqrt{35} (-5 \pi + t)\right] + \sqrt{35} \sin\left[\frac{1}{2} \sqrt{35} (-5 \pi + t)\right]\right)\right) \text{UnitStep}[-5 \pi + t] + \left(\frac{1}{9} - \frac{1}{315} e^{\frac{1}{2}(2 \pi - t)} \left(35 \cos\left[\frac{1}{2} \sqrt{35} (-2 \pi + t)\right] + \sqrt{35} \sin\left[\frac{1}{2} \sqrt{35} (-2 \pi + t)\right]\right)\right) \text{UnitStep}[-2 \pi + t]$$

This is pretty complicated. The solution using DSolve is even worse.

y[t]/.soln

$$\left\{ -\frac{1}{315} e^{-t/2} \left(-18 \sqrt{35} \sin\left[\frac{\sqrt{35} t}{2}\right] + 36 \sqrt{35} e^{9\pi/2} \cos\left[\frac{\sqrt{35} t}{2}\right] \sin\left[\frac{9\sqrt{35} \pi}{2}\right] \right. \right. \\ \text{UnitStep}[-9\pi + t] - 36 \sqrt{35} e^{9\pi/2} \cos\left[\frac{9\sqrt{35} \pi}{2}\right] \sin\left[\frac{\sqrt{35} t}{2}\right] \text{UnitStep}[-9\pi + t] - \\ 35 e^{5\pi/2} \cos\left[\frac{5\sqrt{35} \pi}{2}\right] \cos\left[\frac{\sqrt{35} t}{2}\right] \text{UnitStep}[-5\pi + t] + \\ 35 e^{t/2} \cos\left[\frac{\sqrt{35} t}{2}\right]^2 \text{UnitStep}[-5\pi + t] + \\ \sqrt{35} e^{5\pi/2} \cos\left[\frac{\sqrt{35} t}{2}\right] \sin\left[\frac{5\sqrt{35} \pi}{2}\right] \text{UnitStep}[-5\pi + t] - \\ \sqrt{35} e^{5\pi/2} \cos\left[\frac{5\sqrt{35} \pi}{2}\right] \sin\left[\frac{\sqrt{35} t}{2}\right] \text{UnitStep}[-5\pi + t] - \\ 35 e^{5\pi/2} \sin\left[\frac{5\sqrt{35} \pi}{2}\right] \sin\left[\frac{\sqrt{35} t}{2}\right] \text{UnitStep}[-5\pi + t] + \\ 35 e^{t/2} \sin\left[\frac{\sqrt{35} t}{2}\right]^2 \text{UnitStep}[-5\pi + t] + 35 e^{\pi} \cos[\sqrt{35} \pi] \cos\left[\frac{\sqrt{35} t}{2}\right] \\ \text{UnitStep}[-2\pi + t] - 35 e^{t/2} \cos\left[\frac{\sqrt{35} t}{2}\right]^2 \text{UnitStep}[-2\pi + t] - \\ \sqrt{35} e^{\pi} \cos\left[\frac{\sqrt{35} t}{2}\right] \sin[\sqrt{35} \pi] \text{UnitStep}[-2\pi + t] + \\ \sqrt{35} e^{\pi} \cos[\sqrt{35} \pi] \sin\left[\frac{\sqrt{35} t}{2}\right] \text{UnitStep}[-2\pi + t] + \\ 35 e^{\pi} \sin[\sqrt{35} \pi] \sin\left[\frac{\sqrt{35} t}{2}\right] \text{UnitStep}[-2\pi + t] - \\ \left. \left. 35 e^{t/2} \sin\left[\frac{\sqrt{35} t}{2}\right]^2 \text{UnitStep}[-2\pi + t] \right) \right\}$$


6.* Use one of the solution formulas in 5 a to make a function, g. Use g to plot the phase plane trajectory and as well as the state space trajectory for the system.

g[t_] := Evaluate[Lsoln]


g[t]

$$\frac{2 e^{-t/2} \sin\left[\frac{\sqrt{35} t}{2}\right]}{\sqrt{35}} + \frac{4 e^{\frac{1}{2}(9\pi-t)} \sin\left[\frac{1}{2} \sqrt{35} (-9\pi + t)\right] \text{UnitStep}[-9\pi + t]}{\sqrt{35}} - \\ \left(\frac{1}{9} - \frac{1}{315} e^{\frac{1}{2}(5\pi-t)} \left(35 \cos\left[\frac{1}{2} \sqrt{35} (-5\pi + t)\right] + \sqrt{35} \sin\left[\frac{1}{2} \sqrt{35} (-5\pi + t)\right] \right) \right) \\ \text{UnitStep}[-5\pi + t] + \\ \left(\frac{1}{9} - \frac{1}{315} e^{\frac{1}{2}(2\pi-t)} \left(35 \cos\left[\frac{1}{2} \sqrt{35} (-2\pi + t)\right] + \sqrt{35} \sin\left[\frac{1}{2} \sqrt{35} (-2\pi + t)\right] \right) \right) \\ \text{UnitStep}[-2\pi + t]$$

```
ParametricPlot[ {g[t],g'[t]}, {t,0,40}, PlotRange->{{-0.4,0.6},{-1.5,2.2}},
AspectRatio->3/2]
```



```
ParametricPlot3D[ {g[t],g'[t],t}, {t,0,40}, AspectRatio->3/2, ViewPoint->{3,
-15, 50}, PlotPoints->300, PlotRange->{{-0.4,0.6},{-1.5,2.2},{0,40}}]
```



7.* Continuing 6. Plot an animation of the phase space trajectory.

```
<<Graphics`Animation`
```

```
Animate[ ParametricPlot[ {g[t],g'[t]}, {t,0,T},  
PlotRange->{{-0.4,0.6},{-1.5,2.2}}, AspectRatio->3/2], {T,1,40}]
```

